


배열 정렬하기과 선택정렬 Selection Sort

배열 안에 있는 요소들을 정렬할 수 있게 됩니다.

소팅(Sorting, 일정한 순서가 되도록 재배열) 절차는 오브젝트^{Object}들이 가진 성질을 서로 비교하고 그 결과에 따라 순서대로 배열하는 절차이다. 배열 소팅은 컴퓨터 과학에서 잘 알려진 영역이고, 이미 많은 배열 소팅 알고리즘들이 있다. 우리는 그 중에서 가장 단순하고 이해하기 쉬운 배열 소팅 알고리즘 하나를 살펴보기로 하자. — 선택 정렬

선택 정렬^{Selection Sort} 알고리즘은 N 개의 요소가 있는 배열 안에서 가장 값이 작은 요소를 찾아서 제일 앞에 있는 요소와 맞바꾼다. 그 다음 나머지 $(N-1)$ 중에서 다시 가장 작은 요소를 찾아서 두 번째 요소와 맞바꾼다. 같은 동작을 $N-2$, $N-3$ 로 계속 해나간다. 결국 요소가 하나만 남게 되면 종료한다. 마지막 남은 요소가 가장 큰 요소이다.

이 알고리즘 실행을 그림으로 보면 다음과 같다. 화살표는 요소를 맞바꾼다는 표시이다.



배열 하나를 오름차순으로 소팅 해보자. 먼저 구현 할 알고리즘 전체를 주석으로 적는다.

```
//배열의 범위 chB와 chE를 입력한다.
//chB와 chE 사이(chB, chE를 포함하여) 값으로 배열 A를 채운다.
//배열 A를 첫 번째 memo에 표시한다.
//배열 A를 오름차순으로 소팅(Sorting) 한다.
//배열 A를 두 번째 memo에 표시한다.
```

이제, 주석으로 기록해 둔 알고리즘 전체를 구문으로 적어보자.

```
procedure TfrmArr.btnGoClick(Sender: TObject);
var
    chB, chE: Integer;
    A: Array_n_elements;
begin
    Input(chB, chE); //배열의 범위 chB 와 chE를 입력한다
    FillArr(a, chB, chE); //chB와 chE 사이(chB, chE를 포함하여) 의 무작위 값으로 배열 A를
    채운다
    OutputArray1(A, n); //배열 A를 첫 번째 TMemo에 표시한다
    SortArr(A); //배열 A를 오름차순으로 소팅(Sorting) 한다
    OutputArray2(A, n); //배열 A를 두 번째 TMemo에 표시한다
end;
```

이미 앞의 모듈 17에서 예제로 설명한 것들은 여기에서 중복하여 설명하지 않겠다. 배열 소팅 프로시저만 같이 작성해보자. *TMemo*를 정렬하던 프로시저와 거의 같다.



Tip | 모듈 17에 있는 예제라서 여기에서 설명을 생략한 내용들

배열의 개수 *n*을 상수로 지정하기, 텍스트박스로 받은 값을 범위의 값으로 지정하는 프로시저, 무작위 값으로 배열을 채우는 프로시저, 메모에 배열을 넣어서 표시하는 프로시저

```

procedure SortArr(var a: Array_n_elements);
var
    i: Integer;
begin
    for i:=1 to n-1 do //배열에 마지막 하나가 남을 때까지
        Change(a[i], a[NumMin(a, i)]); //현재 위치에서 배열의 마지막까지의 요소 중에 가장
        작은 요소를 현재 요소와 맞바꾼다.
    end;

```

위의 프로시저에서 사용되는 Change 프로시저를 작성한다. change 프로시저에는 배열의 인덱스번호, 그리고 그 인덱스번호의 위치에 있는 요소에서부터 마지막 요소까지 중에서 가장 값이 작은 요소를 찾아서 반환하는 함수(여기에서는 NumMin)가 있어야 한다.

```

function NumMin(a: Array_n_elements; start: Integer): Integer;
var
    i, m: Integer;
begin
    m:=start;
    for i:=m+1 to n do
        if a[i]<a[m] then
            m:=i;
    NumMin:=m;
end;

procedure Change(var one, two: Integer);
var
    temp: Integer;
begin
    temp:=one;
    one:=two;
    two:=temp;
end;

```

실습

Exercise 1.

키보드로 값의 범위를 받아서 이 범위 안에 있는 정수들로 배열을 채운다.

배열을 메모에 표시한다.

이 배열을 내림차순으로 소팅하고 두 번째 메모에 표시한다.

Exercise 2.

키보드로 값의 범위를 받아서 이 범위 안에 있는 정수들로 배열을 채운다.

배열을 메모에 표시한다.

양수 값을 가지는 모든 요소들을 배열의 앞쪽으로 옮기고, 0 또는 음수 값을 가지는 모든 요소들은 배열의 뒤쪽으로 옮긴다(소팅 즉 정렬하지는 않는다). 이 배열을 두 번째 메모에 표시한다.

Exercise 3.

키보드로 값의 범위를 받아서 이 범위 안에 있는 *Real*들로 배열을 채운다.

배열을 첫 번째 메모에 표시한다.

배열을 반으로 나누고 앞의 반쪽과 뒤의 반쪽을 서로 맞바꾼다. 이 배열을 두 번째 메모에 표시한다.